
* ========================================================================================================== *
* Earmarked Funding and the Control-Performance Trade-Off in International Development Organizations		 *
* Authors: Mirko Heinzel, Ben Cormier & Bernhard Reinsberg                                                   *
* Version: 22.02.2023                                                                                        *
* Stata 16.2                                                                                                 *
* ========================================================================================================== *

*install packages if necessary
*ssc install reghdfe
*ssc install ppmlhdfe
*ssc install blindschemes
*ssc install ivreghdfe
*ssc install ivreg2
*ssc install ftools
*ssc install sum2docx
*ssc install psacalc
*ssc install plausexog
*ssc install estout

cd "location of extracted files"

*******************
* Main Manuscript *
*******************

*Figure 1
use "Heinzel_Cormier_Reinsberg_23_IO_tsgraph.dta"
graph bar ml mbi, over(year) stack scheme(plotplainblind) legend(off)

*Table 1
clear 
use "Heinzel_Cormier_Reinsberg_23_IO.dta"

reghdfe wb_bank_all earmarked , absorb(countryname sector_fe approval_year) vce(cluster country_year)
estimates store m1, title()

reghdfe wb_bank_all earmarked log_amount pilot_project , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m2, title()

ivreghdfe wb_bank_all mean_perf_lendinstr (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m3, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m4, title()

estfe . m1 m2 m3 m4 , labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m1 m2 m3 m4 using mbi1.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table 2

reghdfe log_superyear earmarked , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m5, title()

reghdfe log_superyear earmarked log_amount pilot_project , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m6, title()

ivreghdfe log_superyear mean_lendinstr_lscostpy (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m7, title()

ivreghdfe log_superyear log_amount mean_lendinstr_lscostpy pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m8, title()

estfe . m5 m6 m7 m8 , labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m5 m6 m7 m8 using mbi2.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table 3

reghdfe six_bank_rating earmarked , absorb(donor countryname sector approval_year) vce(cluster country_year)
estimates store m9, title()

reghdfe six_bank_rating earmarked if donor=="AsianDB" , absorb(donor countryname sector approval_year ) vce(cluster country_year)
estimates store m10, title()

reghdfe six_bank_rating earmarked if donor=="AfricanDB" , absorb(donor countryname sector approval_year ) vce(cluster country_year)
estimates store m11, title()

reghdfe six_bank_rating earmarked if donor=="CDB" , noabsorb vce(cluster donor_country)
estimates store m12, title()

reghdfe six_bank_rating earmarked if donor=="IFAD" , absorb(donor countryname sector approval_year ) vce(cluster country_year)
estimates store m13, title()

reghdfe six_bank_rating earmarked if donor=="WB" , absorb(donor countryname sector approval_year ) vce(cluster country_year)
estimates store m14, title()

estfe . m9 m10 m11 m12 m13 m14 , labels(countryname "Country fixed effects" sector "Sector fixed effects" approval_year "Year fixed effects")

esttab m9 m10 m11 m12 m13 m14 using mbi3.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Oster sensitivity

reg wb_bank_all earmarked i.country_fe i.sector_fe i.approval_year , vce(cluster country_year)
psacalc delta earmarked
*-0.33595

reg log_superyear earmarked i.country_fe i.sector_fe i.approval_year , vce(cluster country_year)
psacalc delta earmarked
*0.49638

************
* Appendix *
************

*Figure A1

clear
use "Heinzel_Cormier_Reinsberg_23_IO.dta"
keep if donor=="WB"
gen all_projects=1
keep if wb_bank_all!=. 
drop if sector=="Multi/other"
drop if sector=="NA"
collapse (sum) all_projects earmarked, by(sector)
drop if missing(sector)
egen sum_all=sum(all_projects)
egen sum_em=sum(earmarked)
gen mean_em=earmarked/sum_em
gen mean_all=all_projects/sum_all
gen mean_core=(all_projects-earmarked)/(sum_all-sum_em)
replace sector="ICT" if sector=="Information and Communications Technologies"
graph hbar mean_em mean_core, over(sector) scheme(plotplainblind) 

*Figure A2

clear
use "Heinzel_Cormier_Reinsberg_23_IO.dta"
keep if donor=="WB"
gen all_projects=1
keep if wb_bank_all!=. 
collapse (sum) all_projects earmarked, by(region)
egen sum_all=sum(all_projects)
egen sum_em=sum(earmarked)
gen mean_em=earmarked/sum_em
gen mean_all=all_projects/sum_all
gen mean_core=(all_projects-earmarked)/(sum_all-sum_em)
graph hbar mean_em mean_core, over(region) scheme(plotplainblind) 

*Figure A3

clear
use "Heinzel_Cormier_Reinsberg_23_IO.dta"
keep if donor=="WB"
gen all_projects=1
keep if wb_bank_all!=. 
collapse (sum) all_projects earmarked, by(income_group)
egen sum_all=sum(all_projects)
egen sum_em=sum(earmarked)
gen mean_em=earmarked/sum_em
gen mean_all=all_projects/sum_all
gen mean_core=(all_projects-earmarked)/(sum_all-sum_em)
gen income_group_fe=1 if income_group=="High income"
replace income_group_fe=4 if income_group=="Low income"
replace income_group_fe=3 if income_group=="Lower middle income"
replace income_group_fe=2 if income_group=="Upper middle income"
labmask income_group_fe, values(income_group)
graph hbar mean_em mean_core, over(income_group_fe) scheme(plotplainblind) 

*Figure A4
clear all
use "Heinzel_Cormier_Reinsberg_23_IO.dta"
histogram wb_bank_all, frac discrete xlabel(1(1)6) graphregion(lcolor(gs10)) graphregion(color(white)) color(gs12) lcolor(black) lwidth(thin)

*Table A1

sum2docx wb_bank_all log_superyear six_bank_rating earmarked log_amount pilot_project mean_em_lendinstr mean_perf_lendinstr  mean_lendinstr_lscostpy if earmarked!=. & six_bank_rating!=. using mbiA1.docx, stats(N mean sd min max) replace

*Table A2

reghdfe earmarked mean_em_lendinstr mean_perf_lendinstr if wb_bank_all!=. , absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m1, title()

reghdfe earmarked mean_em_lendinstr mean_perf_lendinstr log_amount pilot_project if wb_bank_all!=., absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m2, title()

reghdfe earmarked mean_em_lendinstr mean_lendinstr_lscostpy if log_superyear!=., absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m3, title()

reghdfe earmarked mean_em_lendinstr mean_lendinstr_lscostpy log_amount pilot_project if log_superyear!=., absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m4, title()

estfe . m1 m2 m3 m4 , labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m1 m2 m3 m4 using mbiA2.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A3

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb(wb_ieg_evaltype countryname sector_fe approval_year) cluster(country_year) first
estimates store m5, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb(wb_ieg_evaltype countryname sector_year) cluster(country_year) first
estimates store m6, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb(wb_ieg_evaltype country_year sector_year) cluster(country_year) first
estimates store m7, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb(wb_ieg_evaltype country_sector approval_year) cluster(country_year) first
estimates store m8, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb(wb_ieg_evaltype country_year sector_year country_sector) cluster(country_year) first
estimates store m9, title()

estfe . m5 m6 m7 m8 m9  ,  labels(countryname "Country fixed effects" country_year "Country-year fixed effects" country_sector "Country-sector fixed effects" sector_fe "Sector fixed effects" sector_year "Sector-year fixed effects" approval_year "Year fixed effects")

esttab m5 m6 m7 m8 m9 using mbiA3.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A4

reghdfe six_bank_rating earmarked , absorb(donor_year countryname sector) vce(cluster country_year)
estimates store m10, title()

reghdfe six_bank_rating earmarked , absorb(donor_year donor_country sector) vce(cluster country_year)
estimates store m11, title()

reghdfe six_bank_rating earmarked , absorb(donor_year donor_country sector_year) vce(cluster country_year)
estimates store m12, title()

reghdfe six_bank_rating earmarked , absorb(donor_year donor_country country_year) vce(cluster country_year)
estimates store m13, title()
 
estfe . m10 m11 m12 m13 ,  labels(countryname "Country fixed effects" country_year "Country-year fixed effects" country_sector "Country-sector fixed effects" sector_fe "Sector fixed effects" sector_year "Sector-year fixed effects" approval_year "Year fixed effects")

esttab m10 m11 m12 m13 using mbiA4.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A5

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project change_idealdist_us unsc_any (earmarked = mean_em_lendinstr) , absorb( country_year sector_year) cluster(country_year) first
estimates store m14, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project change_corr (earmarked = mean_em_lendinstr) , absorb( country_year sector_year) cluster(country_year) first
estimates store m15, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project change_gdppc (earmarked = mean_em_lendinstr) , absorb( country_year sector_year) cluster(country_year) first
estimates store m16, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project change_gdppc change_corr change_idealdist_us unsc_any (earmarked = mean_em_lendinstr) , absorb( country_year sector_year) cluster(country_year) first
estimates store m17, title()

estfe . m14 m15 m16 m17  ,  labels(countryname "Country fixed effects" country_year "Country-year fixed effects" country_sector "Country-sector fixed effects" sector_fe "Sector fixed effects" sector_year "Sector-year fixed effects" approval_year "Year fixed effects")

esttab m14 m15 m16 m17 using mbiA5.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A6

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(countryname) first
estimates store m18, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_sector) first
estimates store m19, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(sector_year) first
estimates store m20, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(wb_ieg_evalfy) first
estimates store m21, title()

estfe . m18 m19 m20 m21 ,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m18 m19 m20 m21 using mbiA6.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A7

reghdfe wb_bank_all earmarked mean_perf_lendinstr , absorb(countryname approval_year ) vce(cluster country_year)
estimates store m22, title()

reghdfe wb_bank_all earmarked mean_perf_lendinstr , absorb(countryname sector_fe approval_year ) vce(cluster country_year)
estimates store m23, title()

reghdfe log_superyear earmarked mean_lendinstr_lscostpy , absorb( countryname approval_year) vce(cluster country_year)
estimates store m24, title()

reghdfe log_superyear earmarked mean_lendinstr_lscostpy , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m25, title()

estfe . m22 m23 m24 m25 ,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m22 m23 m24 m25 using mbiA7.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A8

reghdfe wb_bank_all earmarked if wb_lendinginstrumenttype=="IPF" & log_superyear!=., absorb(countryname sector_fe approval_year ) vce(cluster country_year)
estimates store m26, title()

reghdfe wb_bank_all earmarked if wb_lendinginstrumenttype=="DPF" & log_superyear!=., absorb(countryname sector_fe approval_year ) vce(cluster country_year)
estimates store m27, title()

reghdfe log_superyear earmarked if wb_lendinginstrumenttype=="IPF" & wb_bank_all!=., absorb(countryname sector_fe approval_year ) vce(cluster country_year)
estimates store m28, title()

reghdfe log_superyear earmarked if wb_lendinginstrumenttype=="DPF" & wb_bank_all!=., absorb(countryname sector_fe approval_year ) vce(cluster country_year)
estimates store m29, title()

estfe . m26 m27 m28 m29 ,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m26 m27 m28 m29 using mbiA8.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A9

ivreghdfe wb_bank_qoe log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m30, title()

ivreghdfe wb_bank_super log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m31, title()

ivreghdfe wb_outcome log_amount mean_outc_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m32, title()

ivreghdfe binary_wb_rating log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m33, title()

ivreghdfe altbinary_wb_rating log_amount mean_perf_lendinstr pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m34, title()

estfe . m30 m31 m32 m33 m34,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m30 m31 m32 m33 m34 using mbiA9.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A10

ologit wb_bank_all earmarked log_amount pilot_project i.country_fe i.sector_fe i.approval_year , vce(cluster country_year )
estimates store m35, title()

logit altbinary_wb_rating earmarked log_amount pilot_project i.country_fe i.sector_fe i.approval_year , vce(cluster country_year )
estimates store m36, title()

logit binary_wb_rating earmarked log_amount pilot_project i.country_fe i.sector_fe i.approval_year , vce(cluster country_year )
estimates store m37, title()

estfe . m35 m36 m37,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m35 m36 m37 using mbiA10.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) pr2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A11

reghdfe wb_bank_all log_count_project log_amount pilot_project  , absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m38, title()

ivreghdfe wb_bank_all log_amount mean_perf_lendinstr pilot_project (log_count_project = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m39, title()

reghdfe wb_bank_all em_main em_additional log_amount pilot_project , absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m40, title()

reghdfe log_superyear log_count_project log_amount pilot_project  , absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m41, title()

ivreghdfe log_superyear log_amount mean_lendinstr_lscostpy pilot_project (log_count_project = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m42, title()

reghdfe log_superyear em_main em_additional log_amount pilot_project  , absorb( countryname sector_fe approval_year) cluster(country_year) 
estimates store m43, title()

estfe . m38 m39 m40 m41 m42 m43,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m38 m39 m40 m41 m42 m43 using mbiA11.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A12

ppmlhdfe wb_supervision_py_2 earmarked log_amount pilot_project , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m44, title()

ppmlhdfe wb_supervision_py_2 earmarked log_amount pilot_project , absorb( country_year sector_year country_sector) vce(cluster country_year)
estimates store m45, title()

ppmlhdfe wb_supervision_2 earmarked log_amount pilot_project , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m46, title()

ppmlhdfe wb_supervision_2 earmarked log_amount pilot_project , absorb( country_year sector_year country_sector) vce(cluster country_year)
estimates store m47, title()

ppmlhdfe wb_completion_2 earmarked log_amount pilot_project , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m48, title()

ppmlhdfe wb_completion_2 earmarked log_amount pilot_project , absorb( country_year sector_year country_sector) vce(cluster country_year)
estimates store m49, title()

estfe . m44 m45 m46 m47 m48 m49, labels(countryname "Country fixed effects" country_year "Country-year fixed effects" country_sector "Country-sector fixed effects" sector_fe "Sector fixed effects" sector_year "Sector-year fixed effects" approval_year "Year fixed effects")

esttab m44 m45 m46 m47 m48 m49 using mbiA12.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) pr2 se mlabels(,titles) l indicate(`r(indicate_fe)')


*Table A13

*5%
plausexog uci wb_bank_all log_amount mean_perf_lendinstr pilot_project i.country_fe i.sector_fe i.approval_year (earmarked = mean_em_lendinstr) , vce(cluster country_year) gmin(-0.0235) gmax(0.0235) level(0.9)

*10%
plausexog uci wb_bank_all log_amount mean_perf_lendinstr pilot_project i.country_fe i.sector_fe i.approval_year (earmarked = mean_em_lendinstr) , vce(cluster country_year) gmin(-0.047) gmax(0.047) level(0.9)

*15%
plausexog uci wb_bank_all log_amount mean_perf_lendinstr pilot_project i.country_fe i.sector_fe i.approval_year (earmarked = mean_em_lendinstr) , vce(cluster country_year) gmin(-0.0705) gmax(0.0705) level(0.9)


*Table A14

*5%
plausexog uci log_superyear log_amount mean_lendinstr_lscostpy pilot_project i.country_fe i.sector_fe i.approval_year (earmarked = mean_em_lendinstr) , vce(cluster country_year) gmin(-0.15) gmax(0.15) level(0.9)

*10%
plausexog uci log_superyear log_amount mean_lendinstr_lscostpy pilot_project i.country_fe i.sector_fe i.approval_year (earmarked = mean_em_lendinstr) , vce(cluster country_year) gmin(-0.31) gmax(0.31) level(0.9)

*15%
plausexog uci log_superyear log_amount mean_lendinstr_lscostpy pilot_project i.country_fe i.sector_fe i.approval_year (earmarked = mean_em_lendinstr) , vce(cluster country_year) gmin(-0.46) gmax(0.46) level(0.9)


*Table A15

ivreghdfe six_bank_rating log_amount mean_perf_lendinstr wb_rec_all pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m50, title()

ivreghdfe six_bank_rating log_amount mean_perf_lendinstr wb_outcome pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m51, title()

ivreghdfe six_bank_rating log_amount mean_perf_lendinstr wb_rec_all wb_outcome pilot_project (earmarked = mean_em_lendinstr) , absorb( countryname sector_fe approval_year) cluster(country_year) first
estimates store m52, title()

estfe . m50 m51 m52,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m50 m51 m52 using mbiA15.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')

*Table A16

reghdfe avgpdo wb_bank_all earmarked num_pdo pdo_table key_matrix comp_analysis , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m53, title()

reghdfe avgpdo wb_bank_all earmarked log_amount pilot_project num_pdo pdo_table key_matrix comp_analysis , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m54, title()

reghdfe avgpdo wb_outcome earmarked num_pdo pdo_table key_matrix comp_analysis , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m55, title()

reghdfe avgpdo wb_outcome earmarked log_amount pilot_project num_pdo pdo_table key_matrix comp_analysis , absorb( countryname sector_fe approval_year) vce(cluster country_year)
estimates store m56, title()

estfe . m53 m54 m55 m56 ,  labels(countryname "Country fixed effects" sector_fe "Sector fixed effects" approval_year "Year fixed effects")

esttab m53 m54 m55 m56 using mbiA16.rtf, star(+ 0.1 * 0.05 ** 0.01 *** 0.001) b(4) r2 se mlabels(,titles) l indicate(`r(indicate_fe)')


